
Solitaire

I have been struggling with this game. I had many ideas about the display of the board, but all
ideas didn't work as wanted. Until I tried this displaymethod. I like how it finally became.

; Solitaire                                                     
; Game 56 in 1K hires for the ZX81                              
                                                                
? * TORNADO *                                                   
                                                                
                                                                
           ORG  #4009             ;#4009                        
           DUMP 49161                                           
                                                                
basic      LD   D,#C0             ; preset for 48K bug          
           JR   init0             ; this game has no 48K bug    
                                                                
           DEFB 236,212,28        ; The BASIC                   
           DEFB 126               ; fully placed over sysvar    
           DEFB 143,0,18          ; start to BASIC=#4009        
                                                                
eline      DEFW last              ; needed to load              
chadd      DEFW last-1                                          
xptr       DEFW 0                                               
stkbot     DEFW last              ; needed to load              
stkend     DEFW last              ; needed to load              
berg       DEFB 0                                               
mem        DEFW 0                                               
           DEFB 0                 ; 128                         
                                                                
init1      JP   init              ; init can be anywhere        
                                                                



; all above reusable AFTER loading                              
                                                                
lastk      DEFB 255,255,255       ; used by ZX81                
margin     DEFB 55                ; used by ZX81                
nxtlin     DEFW basic             ; reusable after load         
                                                                
init0      LD   E,L               ; delay intrupts by           
           DEFB #26               ; LD H,64                     
flagx      DEFB 64                ; clever setting of flags     
                                                                
           XOR  A                 ; intruptcounter reset        
           EX   AF,AF'                                          
                                                                
taddr      DEFW 0                 ; used by ZX81,no hurting code
           LD   B,4               ; frames is set ok            
                                                                
frames     DEFW #DD*256+1         ; used by ZX81, clever IX set 
coprcc     LD   HL,hr             ; set IX                      
sposn      JR   init1                                           
cdflag     DEFB 64                ; used by zx81                
                                                                
                                                                
screensp   DEFW lbuf+#8000        ; each line has               
           DEFW lbuf+#8000+21     ; their own displaybuffer     
           DEFW lbuf+#8000+42                                   
           DEFW lbuf+#8000+63                                   
           DEFW lbuf+#8000+84                                   
           DEFW lbuf+#8000+105                                  
           DEFW lbuf+#8000+126                                  
           DEFW lbuf+#8000+147                                  
           DEFW lbuf+#8000                                      
                                                                
hr         LD   HL,lowres+#8000   ; the lowres display          
           LD   BC,#221           ; minimum needed              
           LD   A,#1E                                           
           LD   I,A                                             
           LD   A,#FB                                           
           CALL #2B5                                            
                                                                
           LD   B,26              ; outline delay for hires     
hr00       DJNZ hr00                                            
                                                                
           LD   HL,cloop                                        
           NOP                                                  
                                                                
           LD   (savesp+1),SP     ; save current stack          
           LD   SP,screensp       ; use display stack           
                                                                
           LD   A,screendata/256                                
           LD   I,A                                             
                                                                
           LD   B,9               ; 9 rows                      
                                                                
bloop      LD   A,B                                             
ypos       XOR  5                 ; test cursor line            
           POP  IX                ; get linebuffer              
                                                                
           NOP                                                  
           LD   A,empline*256/256                               
           JP   Z,showline+#8000  ; in upmem show cursor        
showline   RET  C                 ; in lowmem same timing nop   
           LD   R,A                                             
xpos       DEFW 0,0,0,0,0,0,0                                   
           JP   lowline                                         



                                                                
lowline    LD   E,19              ; E must be 18                
           DEC  DE                ; but I need 6 tstates delay  
           LD   A,(HL)            ; more delay                  
           LD   A,(HL)                                          
           LD   A,(HL)                                          
           LD   A,screendata*256/256                            
                                                                
nline      LD   C,4               ; delay                       
           LD   R,A                                             
nl1        DEC  C                                               
           JR   NZ,nl1                                          
           JP   (IX)              ; do hires display            
                                                                
cloop      CP   (HL)              ; delay                       
           LD   R,A                                             
                                                                
           ADD  A,E               ; go to next line of balls    
           CP   128               ; end reached                 
           JR   C,nline           ; if not next line            
           LD   E,255-17          ; no set for previous lines   
           LD   A,110             ; we need all lines again                     
           NOP                                                  
           LD   C,3                                             
           JR   Z,nl1             ; sync bottom display         
           DJNZ bloop             ; do all rows                 
                                                                
savesp     LD   SP,0              ; retrieve stack              
                                                                
           CALL #292              ; back from intrupt           
           CALL #220                                            
           LD   IX,hr                                           
           JP   #2A4                                            
                                                                
start      LD   HL,lbuf                                         
           LD   DE,lbuf+21                                      
           LD   BC,21*7                                         
           LDIR                   ; set all fields to black     
                                                                
           LD   DE,fielddata                                    
           LD   HL,lbuf+24                                      
                                                                
f0         LD   A,(DE)            ; get nr of rows              
           LD   C,A                                             
           INC  DE                                              
f2         LD   A,(DE)            ; get rowpattern              
           LD   B,8               ; 8 fields on a row           
f3         RLCA                                                 
           INC  HL                                              
           LD   (HL),B                                          
           RR   (HL)              ; bit pattern to byte         
           RRCA                                                 
           RLCA                                                 
           INC  HL                                              
           LD   (HL),B                                          
           RR   (HL)              ; to double byte in fact      
           DJNZ f3                ; do 8 fields per row         
                                                                
           INC  HL                ; point to start of next row  
           INC  HL                                              
           INC  HL                                              
           INC  HL                                              
           INC  HL                                              
           DEC  C                 ; this pattern for more rows  



           JR   NZ,f2                                           
           INC  DE                                              
           LD   A,(DE)            ; test end reached            
           ADD  A,A                                             
           JR   NC,f0                                           
                                                                
           LD   HL,#4040                                        
           LD   (lbuf+84+10),HL   ; erase mid stone             
                                                                
           LD   BC,#404           ; start of cursor             
                                                                
playloop   LD   HL,xpos                                         
ercur      LD   (HL),0                                          
           INC  HL                                              
           LD   A,(HL)                                          
           CP   195                                             
           JR   NZ,ercur          ; erase old cursor            
                                                                
           LD   A,8                                             
           SUB  B                                               
           LD   (ypos+1),A        ; set line in display         
                                                                
           LD   A,C                                             
           ADD  A,A                                             
           LD   HL,xpos-2                                       
           ADD  A,L                                             
           LD   L,A                                             
           LD   (HL),128          ; set cursor on xpos          
           INC  HL                                              
           LD   (HL),128                                        
                                                                
           PUSH BC                ; save old xy                 
           LD   BC,(lastk)                                      
           LD   A,C                                             
           INC  A                                               
           CALL NZ,#7BD           ; translate keypressed        
           POP  BC                ; get xy                      
           LD   HL,nxtlin         ; your keytable               
           CP   (HL)                                            
           INC  HL                                              
           JR   Z,start           ; restart a game              
           PUSH BC                                              
           CP   (HL)                                            
           INC  HL                                              
           JR   NZ,k2                                           
           DEC  B                                               
k2         CP   (HL)                                            
           INC  HL                                              
           JR   NZ,k3                                           
           INC  C                                               
k3         CP   (HL)                                            
           INC  HL                                              
           JR   NZ,k4                                           
           INC  B                                               
k4         CP   (HL)                                            
           INC  HL                                              
           JR   NZ,k5                                           
           DEC  C                                               
k5         CP   (HL)                                            
           JR   Z,fire                                          
           CALL field                                           
           ADD  A,A                                             
           JR   C,false                                         
           POP  HL                                              



           DEFB 254                                             
false      POP  BC                                              
                                                                
selret     CALL delay                                           
                                                                
           JR   playloop                                        
                                                                
fire       CALL field                                           
           BIT  6,A                                             
           JR   NZ,false          ; do not select empty field   
wdir       LD   BC,(lastk)                                      
           LD   A,C                                             
           INC  A                                               
           CALL NZ,#7BD                                         
           LD   HL,nxtlin                                       
           LD   B,4                                             
fdir       INC  HL                                              
           CP   (HL)                                            
           JR   Z,dirfnd                                        
           DJNZ fdir                                            
           JR   wdir              ; wait for valid direction    
                                                                
dirfnd     LD   HL,dirtab+4                                     
fstr       DEC  HL                                              
           DJNZ fstr              ; hl now dy dx for direction  
           LD   A,(HL)                                          
           INC  HL                                              
           LD   L,(HL)            ; l = dx                      
           LD   H,A               ; h = dy                      
           POP  BC                                              
           PUSH BC                ; save start xy               
           ADD  A,B                                             
           LD   B,A                                             
           LD   A,L                                             
           ADD  A,C                                             
           LD   C,A                                             
           PUSH HL                                              
           CALL field             ; step over field             
           POP  HL                                              
           CP   64                ; 64 empty, 128 out of screen 
           JR   NC,false                                        
           PUSH BC                ; save between xy             
           LD   A,H                                             
           ADD  A,B                                             
           LD   B,A                                             
           LD   A,L                                             
           ADD  A,C                                             
           LD   C,A                                             
           LD   (newbc+1),BC      ; set destination xy          
           CALL field                                           
           POP  DE                ; get between                 
           SUB  64                                              
           JR   NZ,false          ; must be empty               
                                                                
; move is valid                                                 
           LD   (HL),A            ; show pin on end             
           INC  HL                                              
           LD   (HL),A                                          
                                                                
           LD   B,D                                             
           LD   C,E                                             
           CALL field                                           
           LD   (HL),64           ; erase pin between           
           INC  HL                                              



           LD   (HL),64                                         
           POP  BC                                              
           CALL field                                           
           LD   (HL),64           ; erase start pin             
           INC  HL                                              
           LD   (HL),64                                         
                                                                
newbc      LD   BC,0                                            
           JR   selret                                          
                                                                
n          EQU  27                                              
                                                                
lowres     DEFB 118                                             
           DEFB 0,0,0,0,0,0,0,0,0,0,0                           
                                                                
           DEFB "S"-n,"O"-n,"L"-n,"I"-n,"T"-n,"A"-n             
           DEFB "I"-n,"R"-n,"E"-n                               
           DEFB 118                                             
                                                                
space      EQU  #4202-$                                         
           DEFS space                                           
                                                                
screendata DEFB 3,192,3,192,3,192,3,192,3,192,3,192             
           DEFB 3,192,3,192,3,192                               
                                                                
l2         DEFB 12,48,12,48,12,48,12,48,12,48,12,48             
           DEFB 12,48,12,48,12,48                               
                                                                
l3         DEFB 16,8,16,8,16,8,16,8,16,8,16,8                   
           DEFB 16,8,16,8,16,8                                  
                                                                
l4         DEFB 32,4,32,4,32,4,32,4,32,4,32,4                   
           DEFB 32,4,32,4,32,4                                  
                                                                
l5         DEFB 32,4,32,4,32,4,32,4,32,4,32,4                   
           DEFB 32,4,32,4,32,4                                  
                                                                
l6         DEFB 64,2,64,2,64,2,64,2,64,2,64,2                   
           DEFB 64,2,64,2,64,2                                  
                                                                
l7         DEFB 64,2,64,2,64,2,64,2,64,2,64,2                   
           DEFB 64,2,64,2,64,2                                  
                                                                
empline    DEFB 0,0,0,0,0,0,0,0,0,0                             
           DEFB 0,0,0,0,0,0,0,0                                 
                                                                
field      LD   HL,lbuf-21+2                                    
           LD   A,B                                             
           AND  7                                               
           INC  A                                               
           LD   DE,21                                           
addy       ADD  HL,DE                                           
           DEC  A                                               
           JR   NZ,addy                                         
           LD   E,C                                             
           ADD  HL,DE                                           
           ADD  HL,DE                                           
           LD   A,(HL)                                          
           RET                                                  
                                                                
delay      LD   A,251                                           
           LD   HL,frames                                       
           ADD  A,(HL)                                          
wfr        CP   (HL)                                            



           JR   NZ,wfr                                          
           RET                                                  
                                                                
dirtab     DEFB 255,0,1,0,255                                   
;               dy dx y x                                       
;                  dy dx dy dx                                  
; Q P A O                                                       
                                                                
fielddata  DEFB 2,%11000111                                     
           DEFB 3,%00000001                                     
           DEFB 2,%11000111                                     
                                                                
lbuf       LD   R,A                                             
           DEFB 128,128,128,128,128,128                         
           DEFB 128,128,128,128,128,128                         
           DEFB 128,128,128,128,128,128                         
           JP   (HL)                                            
                                                                
init       LDIR                   ; repair 48K bug on cursorline
                                                                
           LD   HL,keytab         ; copy key table              
           LD   DE,nxtlin                                       
           LD   C,6                                             
           LDIR                                                 
           JP   start                                           
                                                                
keytab     DEFB 13,10,25,5,26,1   ; rqpaoz                      
                                                                
vars       DEFB 128                                             
?                                                               
last       EQU  $                                               


